Proof Pearl: Revisiting the Mini-Rubik in Coq

Laurent Théry

Marelle Project INRIA, France
Laurent.Thery@inria.fr

Abstract. The Mini-Rubik is the 2x2x2 version of the famous Rubik's Cube. How many moves are required to solve the 3x3x3 cube is still unknown. The Mini-Rubik, being simpler, is always solvable in a maximum of 11 moves. This is the result that is formalised in this paper. From this formalisation, a solver is also derived inside the Coq prover. This rather simple example illustrates how crucial it is to be able to perform safe computation inside a proof system. It also stresses how recent improvements in the Coq architecture have made this formalisation possible.

1 Introduction

A recent paper [3] has shown that 26 moves are sufficient to solve Rubik's cube. This is the best-known upper bound (the exact value is conjectured to be around 20 moves). It uses some clever approximation of the problem but relies mainly on heavy parallel computations: 8000 CPU hours are needed to get the result. In this paper, we tackle the more elementary Mini-Rubik. Instead of 26 small cubes, the Mini-Rubik is composed of 8 small cubes only. It is a well-known result that it is always solvable in a maximum of 11 moves [1]. This is the result we formalise in this paper.

In Coq [6], there is no native data-structure. All basic types such as integer, boolean, string are tree-like structures built using the standard `Inductive` command. For example, the natural numbers use Peano representation with two constructors `S` and `O`. The natural number 3 is then internally represented as `(S (S (S O)))`. If this is perfectly adequate for proofs (for example, the usual inductive principle is given for free), Peano numbers are useless for computing. With a binary representation, the situation gets slightly better but is still not satisfactory. In [5], a generic mechanism is proposed for associating a dedicated data-structure for computing to the standard one for proving while preserving all the nice properties of the type system. This mechanism is applied to integer arithmetic in the following way. First, a special type `Int31` is defined that contains a single constructor with a list of 31 booleans as arguments. This is the reference data-structure. Then, this type is associated in a straightforward manner to the internal 31-bit Ocaml integers[1]. So, computing within the `Int31` type directly benefits from the processor arithmetic with the corresponding speed-up.

[1] In Ocaml [4], the last bit of a word indicates if it should be interpreted as a value or a pointer, the arithmetic has then only 32 - 1 bits.

For example, the `Int31` type was used in [7] to get the primality of some large numbers using elliptic curves.

In this paper, we are going to use the `Int31` type in a slightly different manner. The Mini-Rubik has 3,674,160 possible configurations. In order to get our final result, we need to visit all these configurations while recording which have already been encountered. It amounts in manipulating subsets of a set of size 3,674,160. We break this big set in small pieces and use the `Int31` type to encode subsets of small sets of size 31. This capability of encoding in a single word small subsets is the key aspect that makes our formalisation work.

The paper is organised as follows. In Section 1, we present a naive formalisation that is used to get the basic properties of the problem. In Section 2, we present a second formalisation whose main concern is memory consumption. It is from this second formalisation that we get the main result. Finally, in Section 3, we give more precise information about our formalisation.

2 Direct Formalisation

To represent the Mini-Rubik inside Coq, we have chosen a model that is particularly well-suited for the computation we need to perform. If we believe that one can relatively easily get convinced that we have modelled the Mini-Rubik faithfully, ultimately we should also provide a more intuitive model and formally prove the equivalence with our model.

In our model, the front-upper-left corner remains fixed. So, a configuration needs to take into consideration 7 small cubes only. Information about a small cube is split in two: its position and its orientation. Small cubes are numbered from 1 to 7. The cube on the left of Figure 1 shows which ordering has been used to label the small cubes: the fixed cube and cubes 1,2, 3 composed the front face. Each small cube has only 3 coloured faces, so there are only 3 possible orientations. Orientations are numbered from 1 to 3. A configuration of the Mini-Rubik is then represented by a constructor `State` with 14 positive numbers as arguments (positive numbers are a binary representation of $\mathbb{N}^*$ in Coq).

```
Inductive state: Set :=
  State (p1 p2 p3 p4 p5 p6 p7 o1 o2 o3 o4 o5 o6 o7: positive).
```

$\mathtt{p}_i$ indicates which small cube is at position i and $\mathtt{o}_i$ gives its orientation. We define the initial orientation in such a way that the initial state of the cube is represented as

```
Definition init_state =  State 1 2 3 4 5 6 7 1 1 1 1 1 1 1.
```

Orientations are the less intuitive part of our model. The cube on the right of Figure 1 tries to explain how orientations work. Orientations are numbered from 1 to 3 following the clockwise order. Each small cube has exactly a face that is up or down. We arbitrary decide that it is the number that is held by this face that is recorded in the configuration. So, from the definition of `init_state`, it follows that initially the top and bottom faces contain only 1.



Fig. 1. Positions and Orientations of the Mini-Rubik

The front-upper-left corner being fixed, there are only three elementary rotations of the cube: right, back and down. We obviously recover the usual 6 rotations by symmetry (for the Mini-Rubik we have the following equivalences left = right, up = down and front = back). Also, half-turns and anti-clockwise rotations are obtained by iterating 2 times, resp. 3 times, the respective elementary rotation. All this is rather standard. For orientations, two functions are used. The `up` function changed orientation in a clockwise manner, i.e. $1 \mapsto 2$, $2 \mapsto 3$ and $3 \mapsto 1$. The `down` function does the anti-clockwise change. Rotations are modelled as functions from `State` to `State`:

```
Definition rright s := match s with
State p1 p2 p3 p4 p5 p6 p7 o1 o2 o3 o4 o5 o6 o7 ⇒
State p2 p5 p3 p1 p4 p6 p7 (up o2) (down o5) o3 (down o1) (up o4) o6 o7
end.
Definition rback s := match s with
State p1 p2 p3 p4 p5 p6 p7 o1 o2 o3 o4 o5 o6 o7 ⇒
State p1 p2 p3 p5 p6 p7 p4 o1 o2 o3 (up o5) (down o6) (up o7) (down o4)
end.
Definition rdown s := match s with
State p1 p2 p3 p4 p5 p6 p7 o1 o2 o3 o4 o5 o6 o7 ⇒
State p1 p3 p6 p4 p2 p5 p7 o1 o3 o6 o4 o2 o5 o7
end.
```

This is a direct encoding of Figure 1: cubes 1-4-5-2 composed the right face, cubes 4-5-6-7 the back face and cubes 2-5-6-3 the bottom face. Note also that our choice to use the top and bottom faces to read the orientations is reflected by the fact that the down rotation does not modify any orientation.

A state is reachable if it can be reached from the initial state using the three elementary rotations. This is easily defined inductively by:

```
Inductive reachable: state → Prop :=
  reach0: reachable init_state
| reachr: ∀s, reachable s → reachable (rright s)
| reachb: ∀s, reachable s → reachable (rback s)
| reachd: ∀s, reachable s → reachable (rdown s).
```

The fact that 11 moves are sufficient to solve the Mini-Rubik is true for the half-turn metric. This means that not only elementary rotations need to be considered but also anti-clockwise rotations and half turns. This is done in the `move` relation:

```
Definition move (s1 s2: state) :=
  s2 = rright s1 ∨ s2 = rright (rright s1) ∨
  s2 = rright (rright (rright s1)) ∨ s2 = rback s1 ∨
  s2 = rback (rback s1) ∨ s2 = rback (rback (rback s1)) ∨
  s2 = rdown s1 ∨ s2 = rdown (rdown s1) ∨
  s2 = rdown (rdown (rdown s1)).
```

Once moves are defined, the reachability in n moves is defined inductively as:

```
Inductive nreachable: nat → state → Prop :=
  nreach0: nreachable 0 init_state
| nreachS: ∀n s1 s2, nreachable n s1 → move s1 s2 → nreachable (S n) s2.
```

We also define the property of being reachable in *less than* n moves and the property of being reachable in *exactly* n moves:

```
Definition nlreachable n s := ∃ m, m ≤ n ∧ nreachable m s.
Definition nsreachable n s :=
  nreachable n s ∧ ∀m, m < n → ¬ nreachable m s.
```

Now, the theorem we want to prove can be expressed as:

```
Lemma reach11: ∀s, reachable s → nlreachable 11 s.
```

Turning this lemma into a computational problem is quite direct. For each n, we are going to compute the states that are reachable in less than n moves and the states that are reachable in exactly n moves. We represent states by a simple list of states. On such a list, the function `in_states` checks if a state belongs to the list. We first define the list of all possible moves

```
Definition movel :=
 rright::rright o rright::rright o rright o rright::
 rback::rback o rback::rback o rback o rback::
 rdown::rdown o rdown::rdown o rdown o rdown:: nil.
```

All the states that are reachable in exactly $n+1$ states are included in the states that are in one move of states that are reachable in exactly n states. This is the basic idea of the algorithm. The function `nexts` does this computation for a single state:

```
Definition nexts (ps: states * states) s :=
  fold_left
    (fun (ps: states * states) f ⇒
      let (states, nstates) := ps in
      let s1 := f s in
      if in_states s1 states then ps else (s1 :: states, s1 :: nstates))
    movel ps.
```

where `fold_left` is the tail recursive version of the usual iterative `fold` function on lists. For each state s_1 that is at one move from `s`, the `nexts` function checks if s_1 has already been visited. If not, it is added to the list of visited states (the first element of the pair) and to the list of the new states (the second element of the pair). Finally, to get the states that are reachable in less than n moves and the states that are reachable in exactly n moves, we just need to iterate the `nexts` function starting from the lists composed of the initial state only:

```
Fixpoint iters_aux n (ps: states * states) :=
  match n with
       0 ⇒ ps
  | S n1 ⇒ let (m,p) := ps in iters_aux n1 (fold_left nexts p (m,nil))
  end.
Definition iters n := iters_aux n (init_state::nil, init_state::nil).
```

It is relatively easy to show that if the second pair returns by `(iters n)` is empty, the Mini-Rubik is solvable in `n-1` moves.

3 Optimising Memory Consumption

The implementation of `iters` is far too naive to let us prove the `reach11` theorem. Computing `(iters 5)`, which involves 12,224 states only, is already impossible inside Coq. Nevertheless, `iters` is useful as a reference implementation with which our optimised version is going to be proved equivalent. What `iters` actually does is to compute the diameter of the Cayley graph of the group generated by the three elementary rotations. As explained in [1], having a compact representation in memory of the graph is mandatory to perform this computation. If we go back to how configurations have been encoded, the values of the 14 arguments of `State` are strongly constrained. First of all, the seven arguments (p_1, ..., p_7) which represent positions must be a permutation of (1, 2, 3, 4, 5, 6, 7). Then, the last seven arguments (o_1, ..., o_7) which represent orientations must have values in {1, 2, 3} and their sum should be equal to 1 modulo 3. These constraints are captured by the predicate `valid_state` and are proved to hold for the initial state and to be preserved by the reachability predicate:

```
Lemma reachable_valid: ∀s, reachable s → valid_state s.
```

Note that this theorem already indicates that there are at most $7!3^6 = 3{,}674{,}160$ configurations ($7!$ is the contribution of the permutations, and the 3^6 corresponds to the fact that the value of the last orientation is determined by the value of the other orientations since their sum modulo 3 has a fixed value). Later, we prove that this is actually the exact number of configurations.

An accurate encoding of permutations of length n should also take into consideration the facts that the first element of the permutation has n possible values, the second element $n-1$ and so on. This is done with the following two functions that manipulate permutations as lists:

```
Fixpoint encode_aux l p :=
  match l with
  | nil ⇒ nil
  | m::l₁ ⇒ (if p < m then m−1 else m)::encode_aux l₁ p
  end.
Fixpoint encode l n:=
  match l, n with
  | m::l₁, (S n₁) ⇒ m:: encode (encode_aux l₁ m) n₁
  | _, _ ⇒ nil
  end.
```

As the recursion is not structural, an extra argument `n` is required in the `encode` function to ensure termination. It bounds the length of the resulting list. With this encoding on a permutation of length n, the i^{th} element is ensured to be in $\{1, 2, \ldots, n-i+1\}$. In particular, the last element is always 1 and can be discarded. If n is the length of the permutation we want to encode, this is done by giving $n-1$ as the extra argument of the `encode` function. For example, if we consider the permutation of the initial configuration, its encoding is computed by `(encode (1::2::3::4::5::6::7::nil) 6)` which evaluates to `1::1::1::1::1::1::nil`. To sum up, the information about position in a state can be encoded by 6 numbers $(\mathtt{p}'_1, \mathtt{p}'_2, \mathtt{p}'_3, \mathtt{p}'_4, \mathtt{p}'_5, \mathtt{p}'_6)$ with $0 < \mathtt{p}'_i \leq 8-i$ and the information about orientation can be encoded by 6 numbers $(\mathtt{o}'_1, \mathtt{o}'_2, \mathtt{o}'_3, \mathtt{o}'_4, \mathtt{o}'_5, \mathtt{o}'_6)$ with $0 < \mathtt{o}'_i \leq 3$.

We use decision trees to represent sets of states. The 12 numbers that encode a state $(\mathtt{p}'_1, \mathtt{p}'_2, \mathtt{p}'_3, \mathtt{p}'_4, \mathtt{p}'_5, \mathtt{p}'_6, \mathtt{o}'_1, \mathtt{o}'_2, \mathtt{o}'_3, \mathtt{o}'_4, \mathtt{o}'_5, \mathtt{o}'_6)$ are used to denote a path to a boolean leaf in the decision tree. If this leaf is `true`, the state is in the set. In Coq, a constructor with n arguments allocates $(n+1)$ 32-bit words. It is then better to have the elements with the largest number of arguments at the bottom of the tree structure. This is why the reordering of the path $(\mathtt{p}'_6, \mathtt{o}'_1, \mathtt{o}'_2, \mathtt{o}'_3, \mathtt{o}'_4, \mathtt{o}'_5, \mathtt{o}'_6, \mathtt{p}'_5, \mathtt{p}'_4, \mathtt{p}'_3, \mathtt{p}'_2, \mathtt{p}'_1)$ is favoured. Furthermore, instead of boolean leaves, we can use elements of the `Int31` type to encode sets of 31 elements with the usual convention that the i^{th} element of the set is present if and only if the i^{th} bit is set to one. In our path, $\mathtt{p}'_2$ has 6 possible values and $\mathtt{p}'_3$ has 5 possible values, this means that the pair $(\mathtt{p}'_2, \mathtt{p}'_3)$ has 30 possible values which can be effectively represented by a single `Int31` element (a single bit is then unused). The actual path that is used is then $(\mathtt{p}'_6, \mathtt{o}'_1, \mathtt{o}'_2, \mathtt{o}'_3, \mathtt{o}'_4, \mathtt{o}'_5, \mathtt{o}'_6, \mathtt{p}'_5, \mathtt{p}'_4, \mathtt{p}'_1, 5(p'_2-1)+p'_3)$. Two functions `encode_state` and `decode_state` are defined to relate states and paths and their composition is proved to be the identity on valid states. With this representation, a set of states requires a maximum of 295,001 32-bit words which means 2.6 bits per configuration.

It is also possible to derive a solver by slightly modifying the `iters` program. This follows from the observation that, given a state `s` that is reachable in n moves, the states which are at one move from `s` are reachable in $n-1$, n, or $n+1$ moves. Out of all these states, a solver just needs to be capable to pick one state that is reachable in $n-1$ move. This can be done by building a table that associates to each state `s` the two-bit value $n \,\mathtt{mod}\, 3$ where n is the number moves

that are necessary to reach `s`. To compute this table, we just need to split in two the states that are reachable in less than n moves to get the two-bit information. For example, the modified version of the function `nexts` for the solver is

```
Definition next2s (ps: (states * states) * states) s pmod:=
  fold_left
    (fun (ps: (state * states) * states) f ⇒
      let ((states1, states2), nstates) := ps in
      let s1 := f s in
      if (in_states s1 states1 || in_states s1 states2) then ps
      else match pmod with
             1 ⇒ ((s1::states1, states2), s1::nstates)
           | 2 ⇒ ((states1, s1::states2), s1::nstates)
           | _ ⇒ ((s1::states1, s1::states2), s1::nstates)
           end)
    movel ps.
```

where `pmod` corresponds to the value $n \bmod 3$.

4 Running the Solver

The complete formalisation is available at

`ftp://ftp-sop.inria.fr/marelle/Laurent.Thery/Rubik.zip`

It is composed of 7000 lines of code: 3000 lines for the naive formalisation, 4000 for the optimised version. On a Pentium 4 with 1 Gigabyte of RAM, getting the `reach11` theorem takes 530 seconds. Most of the time is spent in computing `(iters 12)`. Note that, once this computation has been performed, it can also be used to get another interesting result:

```
Lemma valid11: ∀s, valid_state s → reachable s.
```

This proves that the number of configurations of the Mini-Rubik is exactly 3,674,160. This is done by checking that the first element of the pair computed by `(iters 12)` with the optimised version has all its leaves equal to $2^{30} - 1$.

The solver returns the list of moves in the half-turn metric that leads to the initial state. It uses some co-inductive types, which are evaluated lazily, to get the memoisation of the table that associates each state with its index of reachability modulo 3. So, the first time the solver is called, the table is actually computed:

```
Time Eval compute in solve init_state.
     = nil
Finished transaction in 627. secs (627.397621u,0.053992s)
```

The next invocations are then immediate. For example, we can try to swap two adjacent corners:

```
Time Eval compute in solve (State 2 1 3 4 5 6 7 1 1 1 1 1 1 1).
  = Right::Back^-1::Down^2::Right^-1::Back::Right^-1::Back^-1::Right::Down^2::
```

```
      Right::Back::nil
Finished transaction in 0. secs (0.00100000000009u,0.s)
```

or two opposite corners

```
Time Eval compute in solve (State 7 2 3 4 5 6 1 1 1 1 1 1 1 1).
  = Right::Back^-1::Right^2::Back^-1::Right^-1::Down^-1::Right::Down^2::Back::
      Down^-1::Back::nil
Finished transaction in 0. secs (0.00100000000009u,0.s)
```

5 Conclusions

Proof systems like Coq are not well-suited for dealing with state exploration. This paper shows that it is still possible to model problems of relatively large size like the Mini-Rubik. Note that most of the issues we have addressed here is not specific to theorem proving and can also be found in the model checking community (see for example [8]). The main motivation of our formalisation was memory consumption. Having a certified formalisation that uses 2.6 bits only per configuration is rather satisfactory. The 530 seconds to complete the exploration are less satisfactory but could be largely improved. Many optimisations could be added. To name only two, we could use everywhere `Int31` numbers rather than `positive` numbers, we could also precompute the encoding of all the 5040 permutations of `(1, 2, 3, 4, 5, 6, 7)` rather than using on the fly the `encode` function. Finally, if our decision trees are for the moment ad-hoc for the specific configurations of the Mini-Rubik, deriving a generic library that uses `Int31` to represent large finite sets could be useful for other formalisations.

A natural continuation of this work would be to tackle the full Rubik. Obviously, formalising results like [3] is outside reach but getting simpler bounds like the one of 52 moves [2] seems feasible.

References

1. Gene Cooperman and Larry Finkelstein. New Methods for Using Cayley Graphs in Interconnection Networks. *Discrete Applied Mathematics*, 37-38:95–118, 1992.
2. Alexander H. Frey and David Singmaster. *Handbook of Cubik Math*. Enslow Publishers, 1982.
3. Daniel Kunkle and Gene Cooperman. Twenty-Six Moves Suffice for Rubik's Cube. In *ISSAC '07*, pages 235–242, 2007.
4. Xavier Leroy. Objective Caml. Available at http://pauillac.inria.fr/ocaml/, 1997.
5. Arnaud Spiwack. Efficient Integer Computation in Type Theory. 2007. draft paper.
6. The COQ development team. The Coq Proof Assistant Reference Manual v7.2. Technical Report 255, INRIA, 2002. Available at http://coq.inria.fr/doc.
7. Laurent Théry and Guillaume Hanrot. Primality proving with elliptic curves. In *TPHOLs'07*, volume 4732 of *LNCS*, pages 319–333, 2007.
8. Antti Valmari. What the small Rubik's cube taught me about data structures, information theory, and randomisation. *International Journal for Software Tools Technology*, 8(3):180–194, 2006.